

4.5 Wichtige Excel-Funktionen an Beispielen erklärt

Excel bietet eine große Anzahl sehr mächtiger Funktionen, die in der Lage sind, komplexe Berechnungen durchzuführen. Im Rahmen dieses Abschnitts ist es aber nicht möglich, alle diese Funktionen mit der gebührenden Tiefe und Komplexität zu besprechen. Viele dieser Funktionen können zudem ineinander verschachtelt werden, was den Wirkungskreis noch um ein Vielfaches steigert. Ich werde mich deshalb nur auf die gebräuchlichsten und für die meisten Leser brauchbaren Funktionen beschränken und diese anhand von Beispielen erläutern. Komplexe Verschachtelungen, also eine oder zwei verschiedene Funktionen innerhalb einer anderen, muss ich aus Platzgründen Ihrer eigenen Forschungsarbeit überlassen. Dieser Abschnitt soll Ihnen aber zeigen, welche vielfältigen Möglichkeiten schon die Grundfunktionen bieten.

Einfache Funktionen – SUMME und MITTELWERT


Sie haben die Spenden Ihres gemeinnützigen Vereins säuberlich in eine Excel-Tabelle eingetragen und möchten am Jahresende nun wissen, wie viel Geld Sie pro Monat an Spenden eingenommen haben. Dazu müssen Sie die Werte Ihrer Tabelle summieren. Um diese Werte zu summieren, könnten Sie z. B. in Zelle A21 folgende Formel eingeben:

=A2+A3+A4+A5+A6+A7+A8+ etc.

Das ist nicht nur mühsam, sondern auch höchst überflüssig. Für so etwas hält Excel geeignete Funktionen bereit, etwa die Funktion *SUMME*.

	A	B	C	D	E	F	G	H	I	J	K	L
1	Januar	Februar	März	April	Mai	Juni	Juli	August	September	Oktober	November	Dezember
2	313	170	128	191	27	427	412	366	385	381	209	55
3	111	257	381	300	249	316	293	91	32	18	365	355
4	231	144	134	405	495	354	233	116	110	368	74	379
5	253	241	463	313	196	348	373	63	140	299	324	259
6	359	315	439	476	66	33	309	318	133	263	113	339
7	425	140	213	166	92	482	416	128	184	17	179	306
8	405	83	17	352	456	116	152	337	422	168	63	109
9	466	306	76	3	102	174	154	478	80	72	417	264
10	283	353	33	161	478	25	249	467	345	372	180	309
11	154	135	141	229	330	381	287	340	54	213	129	16
12	397	36	363	149	10	292	147	475	343	34	402	354
13	119	284	153	378	207	354	444	168	452	355	480	29
14	198	496	48	332	500	481	369	109	313	257	332	455
15	5	172	364	300	343	9	449	143	55	279	26	328
16	291	54	80	105	7	232	285	370	434	76	26	45
17	368	103	205	426	475	311	134	92	198	249	212	104
18	298	176	98	347	470	121	217	215	440	487	308	499
19	13	329	39	196	244	189	485	86	417	428	454	387
20	214	149	466	263	241	195	278	102	342	97	19	94

Gehen Sie zunächst in die Zelle, in der das Ergebnis der Summation stehen soll. In unserem Beispiel ist das Zelle A21. Dann wählen Sie die Registerkarte *Start*. In der Gruppe *Bearbeiten* klicken Sie auf das Σ -Symbol (*Summe*).



20	18	446
21	=SUMME(A2:A20)	
22	SUMME(Zahl1; [Zahl2] ...)	
23		

Excel schreibt jetzt

=SUMME(A2:A20)

in die Zelle A21, wobei die beiden Zelladressen A2 und A20 markiert sind. Was heißt das nun aber?

Aufgrund des Funktionsnamens *SUMME* weiß Excel, dass es etwas zu summieren gibt, und durch die Adressangabe A2:A20 weiß Excel, welche Zellen summiert werden sollen. Der Doppelpunkt in der Formel steht einfach für das Wort »bis«. Das Programm wird damit aufgefordert, alle Werte zu summieren, die sich in den Zellen A2 bis A20 befinden, einschließlich A2 und A20.

Excel macht Ihnen also sinnvolle Vorschläge, was summiert werden könnte. Sollten das genau die Werte sein, die summiert werden sollen, können Sie nun bestätigen. Sind es nicht die richtigen Zelladressen, können Sie die richtigen ganz normal markieren und erst dann bestätigen.

Damit hat Excel die Werte der ersten Spalte addiert. Und da Sie in der Formel keine absoluten Zelladressen benutzt haben, können Sie diese Formel nun nach rechts kopieren, um auch die Summe aller anderen Monate zu erhalten.

Wenn Sie nur zwei oder drei Zahlen zu summieren haben

TIPP

Wollen Sie nur zwei oder drei Zahlen summieren, sollten Sie diese einfach mit einem Pluszeichen addieren. Die *SUMME*-Funktion ist eine ziemlich umfangreiche Funktion, und bei sehr großen Tabellen kann das ein gehöriger Zeitfaktor werden.

Wenn Sie statt direkt auf das Σ -Symbol auf das Dreieck rechts neben dem Σ -Symbol klicken, erhalten Sie eine Auswahl weiterer einfacher Funktionen.

Möchten Sie zum Beispiel wissen, was Sie im Durchschnitt pro Monat an Spenden eingenommen haben, können Sie die Funktion *MITTELWERT* benutzen.



Die Handhabung ist ähnlich der *SUMME*-Funktion. Sie klicken in die Zelle, in der das Ergebnis stehen soll, zum Beispiel in Zelle A22.

Klicken Sie dann auf das Dreieck neben dem Σ -Symbol, und wählen Sie die Funktion *MITTELWERT*. Auch hier schlägt Ihnen Excel mögliche Koordinaten vor. Sind das die richtigen, brauchen Sie nur noch mit \leftarrow zu bestätigen. Sind sie es nicht, markieren Sie die richtigen Zellen.

In der gleichen Art und Weise verfahren Sie, wenn Sie den maximalen oder minimalen Wert der monatlichen Spenden wissen möchten. Dazu dienen die Funktionen *MAX* und *MIN*. Das weitere Vorgehen mit diesen beiden Funktionen ist identisch mit dem bisher besprochenen Verfahren.

Die Umsatztabelle

	A	B	C	D	E
1	Verkaufspreis	16,80 €			
2	Stückkosten	9,50 €			
3					
4		Stückzahl	Umsatz	Kosten	Gewinn
5					
6	1. Quartal	35	588,00 €	332,50 €	255,50 €
7	2. Quartal	46	772,80 €	437,00 €	335,80 €
8	3. Quartal	87	1.461,60 €	826,50 €	635,10 €
9	4. Quartal	32	537,60 €	304,00 €	233,60 €
10					
11	Summe	200	3.360,00 €	1.900,00 €	1.460,00 €
12	Mittelwert	50	840,00 €	475,00 €	365,00 €
13	Maximum	87	1.461,60 €	826,50 €	635,10 €
14	Minimum	32	537,60 €	304,00 €	233,60 €
15					

könnte mit folgenden Formeln versehen sein:

	A	B	C	D	E
1	Verkaufspreis	16,8			
2	Stückkosten	9,5			
3					
4		Stückzahl	Umsatz	Kosten	Gewinn
5					
6	1. Quartal	35	=B\$1*B6	=B\$2*B6	=C6-D6
7	2. Quartal	46	=B\$1*B7	=B\$2*B7	=C7-D7
8	3. Quartal	87	=B\$1*B8	=B\$2*B8	=C8-D8
9	4. Quartal	32	=B\$1*B9	=B\$2*B9	=C9-D9
10					
11	Summe	=SUMME(B6:B9)	=SUMME(C6:C9)	=SUMME(D6:D9)	=SUMME(E6:E9)
12	Mittelwert	=MITTELWERT(B6:B9)	=MITTELWERT(C6:C9)	=MITTELWERT(D6:D9)	=MITTELWERT(E6:E9)
13	Maximum	=MAX(B6:B9)	=MAX(C6:C9)	=MAX(D6:D9)	=MAX(E6:E9)
14	Minimum	=MIN(B6:B9)	=MIN(C6:C9)	=MIN(D6:D9)	=MIN(E6:E9)
15					

Funktionen der Kategorie Datum & Zeit

Das Berechnen der Kalenderwoche eines gegebenen Datums

Excel bietet vielfältige Möglichkeiten, in einer Tabelle mit einem Datum umzugehen. Man muss aber ein paar Regeln einhalten, um richtige Ergebnisse zu bekommen, denn eigentlich kennt Excel ein Datum gar nicht.


Excel rechnet im Grunde nur mit Tagen, d. h., es hat keine Ahnung von einem 15. April 2022, sondern weiß nur, wie viele Tage seit einem von den Programmierern als Startpunkt vorgegebenen Datum vergangen sind. Für den Startpunkt haben die Programmierer den 1.1.1900 vorgesehen.

Sie müssten also die Anzahl der Tage nur in Jahre und Wochen umrechnen, um zu einem Datumswert zu kommen. Das hört sich kompliziert an und wäre es auch, wenn diese Umrechnung nicht von Excel selbsttätig durchgeführt würde.

Eine der Funktionen dazu ist die Funktion *KALENDERWOCHE*. Nehmen wir einmal an, Sie möchten wissen, in welche Kalenderwoche der 15. April 2022 fällt. Schreiben Sie das Datum in die Zelle A1 einer Tabelle, und gehen Sie dann in die Zelle, in der das Ergebnis stehen soll. Wählen Sie nun auf der Registerkarte *Start* in der Gruppe *Bearbeiten* das kleine Dreieck beim Σ -Symbol. Hier wählen Sie *Weitere Funktionen*.

Sie erhalten ein Fenster mit allen Funktionen, die Excel Ihnen bietet. Wählen Sie bei *Kategorie auswählen* hier *Datum & Zeit*.

Nun wählen Sie *KALENDERWOCHE* und bestätigen die Auswahl mit *OK*.



Im folgenden Fenster erwartet Excel von Ihnen die fortlaufende Zahl seit dem 1.1.1900. Keine Angst, Sie müssen jetzt nicht die Tage vom 1.1.1900 bis zum 15. April 2022 zählen. Das macht Excel automatisch. Sie brauchen nur durch Klick die Zelle auszuwählen, in der Ihr Datum steht, oder Sie können die Zelladresse auch per Tastatur hineinschreiben. Bestätigen Sie zunächst einmal das Ganze mit OK. Sie sollten nun die Zahl 16 erhalten haben, denn der 15. April 2022 liegt in der 16. Kalenderwoche des Jahres 2022. Achtung! Gezählt werden die amerikanischen Kalenderwochen, die Woche mit dem 1. Januar wird dort stets mitgerechnet, sodass es in diesem Fall eine Abweichung zur deutschen Kalenderwoche gibt (in Deutschland fällt der 15. April 2022 in die 15. Kalenderwoche).




Im Fenster der *KALENDERWOCHE*-Funktion gab es noch eine weitere Möglichkeit zur Eingabe, nämlich *Zahl_Typ*. Und *Zahl_Typ* kann nur die Werte 1 und 2 annehmen.

Was hat es damit auf sich? Ganz einfach: Damit legen Sie fest, ob eine Kalenderwoche an einem Sonntag oder an einem Montag beginnen soll. Geben Sie einmal in Zelle A2 das Datum 21. April 2022 ein. Excel sollte Ihnen nun sofort die Zahl 17 anzeigen, denn das Datum liegt in der 17. Kalenderwoche im Jahr 2022.

Nun doppelklicken Sie als Nächstes auf die Zelle B2, um die *KALENDERWOCHE*-Funktion zum nachträglichen Bearbeiten zu öffnen.

Klicken Sie nun zwischen Zelladresse A2 und die schließende Klammer), und geben Sie dort ;2 ein. Damit legen Sie fest, dass die Kalenderwoche an einem Montag beginnt. Im Menü, das sich öffnet, finden Sie aber auch noch weitere Optionen für den Start der Kalenderwoche. Das Semikolon ist wichtig, denn damit trennen Sie mehrere Angaben innerhalb einer Funktion.



In der Kategorie *Datum* kennt Excel noch einige weitere brauchbare Funktionen. Da diese recht einfach in der Handhabung sind, möchte ich sie Ihnen nur in einer Tabelle vorstellen.

Wenn in Zelle A1 das Datum 12.05.2022 steht, führt die Funktion zu folgendem Ergebnis
=TAG(A1)	12
=MONAT(A1)	5
=JAHR(A1)	2022
=WOCHENTAG(A1)	1
=HEUTE()	aktuelles Datum

Die Funktion *WOCHENTAG* gibt Ihnen aber nicht den Wochentag als Wort aus, sondern als Zahl, d. h., ein Sonntag ergibt die Zahl 1, Montag = 2, Dienstag = 3 etc.

Die Wochenenden sollen rot werden


Sie haben in Excel einen Terminkalender angelegt und möchten, dass die Wochenenden durch eine andere Färbung hervorgehoben werden. Dazu können Sie natürlich die bedingte Formatierung nehmen, wenn Sie noch über ein paar zusätzliche Informationen verfügen.

In der Spalte A einer Tabelle wurde der Anfangswert 1. Mai 2022 eingetragen und nach unten kopiert. Da Sie natürlich nicht nur das Datum wissen möchten, sondern auch den Wochentag, wurde Spalte A mit diesem selbst erstellten Format versehen:

TTTT, TT. MMMM JJJJ

Sollten Sie nicht mehr genau wissen, wie solche Formate erstellt werden, lesen Sie das im Abschnitt über Formatierungen weiter oben noch einmal nach.


Ziel ist es, dass für alle Samstage und Sonntage die Zellen von Spalte A rot eingefärbt werden. Dazu brauchen Sie die Möglichkeiten der bedingten Formatierung. Markieren Sie die Zellen A1 bis A31 (wie in der Abbildung zu sehen), und gehen Sie über die Registerkarte *Start* in die Gruppe *Formatvorlage*. Dort wählen Sie dann *Bedingte Formatierung/Regeln zum Hervorheben von Zellen/Weitere Regeln*.



Im nun folgenden Fenster wählen Sie als Regeltyp *Formel zur Ermittlung der zu formatierenden Zellen verwenden*. Nun tragen Sie in das entsprechende Feld folgende Formel ein:

=WOCHENTAG(\$A1)=7

und legen über *Formatieren* eine Füllfarbe fest.



Nachdem Sie dann alle Fenster mit *OK* bestätigt haben, können Sie sich das Ergebnis anschauen. Es wurden aber nur alle Samstage rot hervorgehoben, keine Sonntage. Das Problem werden wir gleich lösen.

Zunächst möchte ich die Frage beantworten, was die Formel =WOCHENTAG(\$A1)=7 eigentlich macht. Sie wissen, dass die Funktion WOCHENTAG aus einem Datum den Wochentag ermittelt, diesen aber nicht als Wort, sondern als Zahl ausgibt. Die Zahl 7 steht für den Samstag. Sie sagen Excel dadurch also, dass Sie einen Samstag suchen und die entsprechenden Zellen rot färben wollen. In der folgenden Tabelle sehen Sie die Werte für alle sieben Wochentage.

Wochentag	Zahl
Sonntag	1
Montag	2
Dienstag	3
Mittwoch	4
Donnerstag	5
Freitag	6
Samstag	7

Möchten Sie also auch die Sonntage rot einfärben, verwenden Sie in einem zweiten Durchlauf die Formel


=WOCHENTAG(\$A1)=1

denn bei Excel beginnt die Woche mit dem Sonntag (= 1).

Das Ganze lässt sich aber auch in nur einem Schritt ausführen. Dabei müssen Sie Excel allerdings sagen, dass nun zwei Bedingungen zutreffen müssen, nämlich $WOCHENTAG(\$A1)=7$ und $WOCHENTAG(\$A1)=1$. Und Sie müssen Excel mitteilen, ob diese beiden Bedingungen mit einem ODER oder mit einem UND verbunden werden sollen, also ob beide Bedingungen gleichzeitig zutreffen müssen (UND) oder ob eine Bedingung genügt (ODER), um die Zellen rot zu färben.

In unserem Beispiel können nicht beide Bedingungen gleichzeitig zutreffen, denn in besagter Zelle A1 steht entweder Samstag oder Sonntag. Ein bestimmtes Datum kann nicht Samstag und Sonntag zugleich sein.

Die Formel, die Sie benutzen müssen, sehen Sie in folgender Abbildung:



Der logische Verknüpfungsoperator ODER hat allgemein folgendes Aussehen:

=ODER(Bedingung_1;Bedingung_2;...;Bedingung_n)

Wir werden uns den logischen Operatoren UND und ODER noch sehr ausführlich im Abschnitt über logische Funktionen ab Seite 360 zuwenden.

Die Wochenenden sollen gar nicht erst auftauchen

Wenn Sie die Wochenenden erst gar nicht sehen wollen, bietet Excel die Möglichkeit, diese gleich ganz aus der Datenliste zu entfernen.

Geben Sie in eine Zelle das Anfangsdatum ein, und kopieren Sie alles ganz klassisch nach unten. Klicken Sie dann in der nach unten kopierten Tabelle unten rechts auf das Symbol, wählen Sie dort *Arbeitstage ausfüllen*. Damit entfernt Excel alle Wochenenden aus Ihrer Liste.



Wie lassen sich die Monate hochzählen?

Vielleicht machen Sie Ihre Abrechnungen immer am gleichen Tag im Monat, also zum Beispiel am 15.3.2022, 15.4.2022, 15.5.2022 etc. Damit haben Sie in Excel ein kleines Problem. Excel zählt zwar ganz einfach die Tage hoch, aber die Monate hochzählen zu lassen funktioniert nur mit einer speziellen Funktion. Das liegt daran, dass Excel intern nur mit Tagen arbeitet. Jedes Mal also, wenn ein Datum benötigt wird, wird eine Zahl wie 44635 in ein uns ge-läufiges Datumsformat umgerechnet.

Aber glücklicherweise gibt es in Excel einige Möglichkeiten, die Ihnen hier helfen können. Eine davon ist, mithilfe der Funktionen *TAG*, *MONAT*, *JAHR* die einzelnen Teile des Datums zu separieren, dann den Monat um 1 zu erhöhen, um anschließend alles wieder zu einem Datum zusammensetzen. Das alles kann man mit einer einzigen Formel schaffen. Aber Excel kennt etwas viel Besseres, eine spezielle Funktion nämlich. Diese Funktion heißt *EDATUM*.

EDATUM hat im Allgemeinen folgende Syntax:

=EDATUM(Ausgangsdatum;Monate)

Wenn also in Zelle A1 das Datum 15.3.2022 steht, gehört in Zelle A2 folgende Formel:

=EDATUM(A1;1)

wenn immer nur ein Monat weitergezählt werden soll.